
News 
The latest version of Spike2 is 5.12, which is available to download from the C
registered v5 users. This revision includes changes to eliminate the flickering
frequently updated windows, for example during sampling. Process schedulin
improved to stop Spike2 taking a lot of the CPU time during sampling, and a s
added to the Preferences dialog to control how Spike2 allocates time for back

A full list of new features is available from the updates section of the website.

A demonstration version of the current software is also available here. 

Version 4.22 of Spike2 is downloadable for registered v4 users. 

Version 3.03 of Signal is now available. This revision includes a frame subset
operations and memory view processing. This allows specific groups of frame
processed. 

A demonstration version of Signal version 3 is available here. 

Version 2.15 of Signal is downloadable for registered v2 users. 

 

We will be exhibiting at the following forthcoming meetings: 

•  International Society of Autonomic Neuroscience: Marseille, France, 

•  Physiological Society Meeting: Bristol, U.K., 20th – 23rd July 

If you are attending either meeting, please feel free to stop by our booth.  

Spike2  
Q. How do I go about playing two (or more) waveforms simultaneously f

A. In order to play out simultaneous waveforms from the 1401, the selec
specified in a single ‘Play wave’ area of the on-line play waveform list
copied to the 1401 memory when you open a new data file for sampli
wave areas that can be copied to the 1401 depends on their size and
using. Spike2 always reserves 256kB of 1401 memory for sampling s
available 1401 memory to store waveforms for replay. 1401 memory 
sampling using the script language to give greater flexibility. 

Output waveform dialog 

 The simplest way to load waveforms into the Play wave list is by selecting channels in a data file 
(holding down Ctrl and clicking on the channel numbers) and using the Output Waveform… 
command from the Sample menu in Spike2. Each play wave area will hold up to 4 data channels 
for output and the time range of data and the DAC outputs to use are selected from the dialog. 
Clicking the Add to online button will set aside memory to hold the selected channels and the label 
and key code that are used to define the play wave area are set here. Further options for play 
wave output such as number of cycles to play, the output speed and whether the play wave area 
is to be triggered (from the 1401 Trigger or E3 input) or linked to other play wave areas defined in 
the list, are available from the Play waveform tab of the sampling configuration, as below. 

  



 

 When you open a new data file ready to sample, any play wave areas are copied to the 1401 
memory ready for output. There are several ways to start output of a wave during sampling; use 
the associated button in the Play waveform toolbar, or press the associated key code to start the 
output manually. Waveform output can also be triggered from a script (using the SampleKey() 
command), or from the output sequencer (using the WAVEGO command). See the Spike2 help 
pages or manual for further information about interactive and scripted arbitrary waveform output.  

   

Signal 

Q. It would be useful to be able to use reference times when using the Time shift data channel 
modification rather than working out the number of data points needed to shift the data by the 
desired amount. Are there any plans to change this? 

A. The latest version of Signal includes changes to the Shift data.. option which is now done 
by specifying a reference time and a new time. This means that, for example, data can easily be 
time shifted relative to the position of a Cursor. 

     

Scripts: Spike2  

Q. Is there any way to print out information about the current sampling configuration? Something like 
a report text file holding information such as number and type of channels, sample rates and 
defined sequencer outputs, for example. 

A. The attached script, PrintConfig.s2s,          prints the current sampling configuration information  
to the Log view and displays it on the screen. You must have a 1401 connected to get channel 
information before running the script. This script is currently for version 5 of Spike2 only, but could 
be modified for earlier versions of Spike2 if needed. Please contact us if you need the script 
modified for earlier versions. 

 

Scripts: Signal  

Q. I would like a way to generate customised text output of measurements where I can add 
information such as the file name, time range of data and specify the text format. I know that you 
can use the Trend Plot command and then copy the results as text, but this only gives you the 
information from the XY-view and there is no way to include information about the data file that 
was measured.  

A. The attached script, DoRMS.sgs,         is an example of customised text output to the Log view. 
The script prompts you to open a data file and then presents a dialog that allows selection of the
time range to measure and either all frames or a frame subset. The script measures the RMS 
amplitude from visible waveform channels in the nominated frames, and prints out the results for 
each channel in columns ready for export to a spreadsheet package. It includes the filename and 
the selected start and end time for measurements along with the RMS values. You could easily 
modify the script to include additional information such as frame states, for example. 

Did you know…? 
The latest versions of Spike2 and Signal now attempt to stop a system hibernating or starting screen 
savers while sampling or running a script. They will also save any sampling data file if there is a battery 
low condition or it receives a system powering down now message. 


'$PrintConfig|Script to printout the current sampling configuration.
'To get channel information we need to have a 1401 connected
'SG 10/05/05

Var Log%;
Var ver;

Log%:=Loghandle();	'Get the name of the log view and then
View(log%);
Window(0,0,50,100);	'Size it
FrontView(log%);
EditselectAll();		'Clear any exisiting text
EditClear();

General%();			'Functions to jump to
Channels%();
Resolution%();
Mode%();
Sequencer%();
PlayWave%();
Automation%();

Func General%();
'Var ver;
Var system%;
 Printlog("General Information");
 Printlog("\n %s",Date$());	'Print the date
 Ver:=App(-1);
 Ver:=Ver/100;
 Printlog("\n Spike2 version number %.2f", Ver);'Print the version of Spike2 
 Printlog("\n Spike2 serial number %d", App(-3));	'and the serial number

 System%:=System();	'Get the operating system we are using

Docase	'Print the OS
	Case System%=351 then
		Printlog("\n Operating System = NT 3.51");
	Case System%=400 then
		Printlog("\n Operating System = Windows 95 or NT");
	Case System%=410 then
		Printlog("\n Operating System = Windows 98");
	Case System%=490 then
		Printlog("\n Operating System = Windows Me");
	Case System%=500 then
		Printlog("\n Operating System = NT 2000");
	Case System%=501 then
		Printlog("\n Operating System = Windows XP");
Endcase

Return 1
End;

Func Channels%();
Var Data%;
Var List%[101];
Var j%;
Var Kind%;
Var Pre%, Trace%;

Printlog("\n\nChannel information");	
	Printlog("\n Maximum permitted number of channels %d", SampleChannels());	'Max number of channels in a new file 

Data%:=FileNew(0);

	If Data%<0 then	'If we cannot access a 1401
		Printlog("\n Channel information unavailable because %s", Error$(Data%)); 'Print why we can't
		Return 1;
	Endif
	Printlog("\n#\tType\tPort\tTitle\tEvt Rate\tADC Rate\tScale\tOffset\tUnits\tPoints\tPre\tTraces\tComment"); 'Print the column titles
	View(Data%);
	ChanList(List%[]);	'Get the list of channels in the open data file
	For j%:= 1 to List%[0] do
	Printlog("\n%d", List%[j%]);	'Print channel number

	Kind%:=ChanKind(List%[j%]);	'Find out what type the channel is
		
	 Docase	'Depending on what type print the relavent information
	  Case Kind%= 1 then
	  	Printlog("\tWaveform");
		Printlog("\t%d",ChanPort(List%[j%]));
		Printlog("\t%s",Chantitle$(List%[j%]));
		Printlog("\t\t%.3f", 1/binsize(List%[j%]));
		Printlog("\t%.2f", Chanscale(List%[j%]));
		Printlog("\t%.2f", Chanoffset(List%[j%]));
		Printlog("\t%s", Chanunits$(List%[j%]));
		Printlog("\t\t\t\t%s", Chancomment$(List%[j%]));
	  Case kind%=2 then
	   Printlog("\tEvent-");
		Printlog("\t%d",ChanPort(List%[j%]));
		Printlog("\t%s",Chantitle$(List%[j%]));
		Printlog("\t\t\t\t\t\t\t\t\t%s", Chancomment$(List%[j%]));
	  Case kind%=3 then
		Printlog("\tEvent+");
		Printlog("\t%d",ChanPort(List%[j%]));
		Printlog("\t%s",Chantitle$(List%[j%]));
		Printlog("\t\t\t\t\t\t\t\t\t%s", Chancomment$(List%[j%]));
	  Case kind%=4 then
		Printlog("\tLevel");
		Printlog("\t%d",ChanPort(List%[j%]));
		Printlog("\t%s",Chantitle$(List%[j%]));
		Printlog("\t\t\t\t\t\t\t\t\t%s", Chancomment$(List%[j%]));
	  Case kind%=5 then
	   Printlog("\tMarker");
		Printlog("\t\t%s",Chantitle$(List%[j%]));
		Printlog("\t\t\t\t\t\t\t\t\t%s", Chancomment$(List%[j%]));
		'Printlog("\n  Number of reals %d", MarkInfo(List%[j%]));
	  Case kind%=6 then
		Printlog("\tWaveMark");
		Printlog("\t%d",ChanPort(List%[j%]));
		Printlog("\t%s",Chantitle$(List%[j%]));
		Printlog("\t");
		Printlog("\t%.3f", 1/binsize(List%[j%]));
		Printlog("\t%.2f", Chanscale(List%[j%]));
		Printlog("\t%.2f", Chanoffset(List%[j%]));
		PrintLog("\t%s", ChanUnits$(List%[j%]));
		Printlog("\t%d", MarkInfo(List%[j%],Pre%, Trace%));
		Printlog("\t%d", Pre%);
		Printlog("\t%d", trace%);
		Printlog("\t%s", Chancomment$(List%[j%]));

	  Case kind%=8 then
		Printlog("\tTextMark");
		Printlog("\t%d",ChanPort(List%[j%]));
		Printlog("\t%s",Chantitle$(List%[j%]));
		Printlog("\t\t\t\t\t\t\t\t\t%s", Chancomment$(List%[j%]));
		'Printlog("\n  Number of characters %d", MarkInfo(List%[j%]));
	 Endcase
	Next;
FileClose(0,-1);
Return 1
end

Func Resolution%();	'Print out the resolution tab information
Var opt%, group%, type%;
Printlog("\n\nResolution settings");
	Printlog("\n Micro seconds per time unit %d uS", SampleUsPerTime()); 'Print the mS/time and time units per ADC
		Printlog("\n Time units per ADC convert %d", SampleTimePerADC());

Opt%:=SampleOptimise(0);	'What optimisation?
	Docase
		Case Opt%=0 then
			Printlog("\n No optimise of microseconds per time or time per ADC.");
		Case Opt%=1 then
			Printlog("\n Partial, optimise time per ADC, no optimise of microseconds per time.");
		Case Opt%=2 then
			Printlog("\n Full, optimise both time per ADC and microseconds per time.");
	Endcase

Group%:=SampleOptimise(1);	'Quick channels?
	Docase
		Case Group%=0 then
			Printlog("\n Version 3 compatible; no Quick channels and channel divides up to 65535.");
		Case Group%=1 then
			Printlog("\n Group channels with the same ideal rate so they get the same actual rate.");
		Case Group%=2 then
			Printlog("\n Optimise for the least error in rate; channels with the same ideal rate may get different actual rates.");
	Endcase

Type%:=SampleOptimise(2); 'Get the 1401 type

	Docase
		Case Type%=0 then
			Printlog("\n 1401 type is set for working with all supported 1401s except a 1401plus with an old ADC.");
		Case Type%=1 then
			Printlog("\n 1401 type is 1401plus with an old ADC");
 		Case Type%=2 then
			Printlog("\n 1401 type is Power1401");
		Case Type%=3 then
			Printlog("\n 1401 type is 1401plus or a micro1401");
		Case Type%=4 then
			Printlog("\n 1401 type is Micro1401 mk II");
		Case Type%=5 then
			Printlog("\n 1401 type is Power1401 625");
		Case Type%=6 then
			Printlog("\n 1401 type is micro1401 (not mk II)");
 Endcase

 Printlog("\n Low limit for microseconds per time = %d uS", SampleOptimise(3)); 'Low and high limits for optimisation
 Printlog("\n High limit for microseconds per time = %d uS", SampleOptimise(4));
Return 1
end

Func Mode%();	'The sampling mode. Continous, timed or triggered.
var Mode%;
Var i%;
Var list%;
Var Code%;

Printlog("\n\nSampling Mode");
Mode%:=SampleMode(0); 'Get the sampling mode

	Docase	'Print what it is
		Case Mode%=1 then
			Printlog("\n Continuous");				
		Case Mode%=2 then
			Printlog("\n Timed");
			Printlog("\n For %d S", SampleMode(-1)); 'Pritn the times
			Printlog("\n Every %d S", SampleMode(-2)); 'Does not tell us whether minutes are used
		Case Mode%=3 then
			Printlog("\n Triggered");
			Printlog("\n For %d S", SampleMode(-1)); 'For in triggered mode doesn't make sense anymore
			Printlog("\n Trigger channel ", Samplemode(-3));
			For i% := 1 to 4 do
				Printlog("\n Trigger %d set-up",i%);
					If sampleTrigger(i%,-1) > 0 then
					Printlog("\n Trigger channel %d",SampleTrigger(i%,-1));
					Code%:=SampleTrigger(i%,-2);
						If Code% =-1 then
							Printlog("\n Trigger code : Accept all");
						 Else
							Printlog("\n Trigger code : %s %02X",Chr$(SampleTrigger(i%,-2)),SampleTrigger(i%,-2));	
						Endif
					Printlog("\n Start time %f S",SampleTrigger(i%,-3));
					Printlog("\n End time %f S", SampleTrigger(i%,-4));
					List%:=SampleTrigger(i%,-5);
					DoCase
						case list%=-1 then
							Printlog("\n All channels");
						case list%=-1 then
							Printlog("\n One channel");
						case list%=-1 then
							Printlog("\n Multiple channels\n");
					Endcase
					Endif
			 Next
	Endcase

Return 1
end

Func Sequencer%();
Var Seq$;
Var SeqStep%;

Seq$:=SampleSequencer$(); 'Find out if we have an output sequencer file

Printlog("\n\nOutput sequencer");

If seq$="" then
		Printlog("\n None or Graphical Sequencer used"); 'Can't find out of graphical seq used
	else
		Printlog("\n %s", Seq$);	'Print the name of the file
		
		SeqStep%:=SampleSeqCtrl(1); 'How is the sequencer controlled
			Printlog("\n Sequencer controlled by:");
		Docase
			Case seqStep%= 0 then
				Printlog(" Keyboard, control panel and script");
			Case seqStep%= 1 then
				Printlog(" Control panel and script");
			Case seqStep%= 2 then
				Printlog(" Script only");
		Endcase
Endif;

Return 1
end

Func Playwave%();
Var List$;
Var Key$;
Var K%;
Var num%;
Var trig%;

List$:=PlayWaveInfo$();	'Any arbitrary outputs used?
If List$ > "" then
	PrintLog("\n\n\Playwaves");

For k%:= 1 to Len(List$) do;
	Key$:=Left$(List$,1);
	
	num%:=PlayWaveChans(Key$);
Var Ch%[Num%];
	PlayWaveChans(Key$,Ch%[]);
	PrintLog("\n %d. Name %s. Key %s. DAC %d, Rate %.3f.", K%,PlaywaveLabel$(Key$),Key$,cH%[],PlayWaveRate(Key$));

	If PlayWaveLink$(Key$)>"" then
	  Printlog(" Linked to %s", PlayWaveLink$(Key$));
	endif
	Trig%:=PlaywaveTrigger(Key$);
		Docase
			Case Trig%=0 then
				Printlog("\n Triggered = NO.");
			Case Trig%=1 then
				Printlog("\n Triggered = YES.");
			Case Trig%=-1 then
				Printlog("\n Triggered ERROR.");
		Endcase

	Printlog(" Play wave speed %.2f.",PlaywaveSpeed(Key$));
	Printlog(" Cycles %d", PlayWaveCycles(Key$));
   LIST$:=DelStr$(List$,1,1);
Next;
Endif;
Return 1
end

Func Automation%();	'Printing the automation tab
Var Size%, Time%;
Printlog("\n\nAutomation settings");
Size%:=SamplelimitSize();
		If Size%< 0 then
			Printlog("\n Stop at file size disabled");
		Else
			Printlog("\n Stop file when %d kB reached",Size%);
		Endif
Time%:=SampleLimitTime();
		If Time%< 0 then
			Printlog("\n Stop at time disabled");
		Else
			Printlog("\n Stop file when %d seconds reached",Time%);
		Endif

	Printlog("\n Number of repeats %d",SampleRepeats());
	If FilePath$(3)>"" then
		Printlog("\n Automatic file path: %s", FilePath$(3));
	Else
		Printlog("\n Automatic file path: None");
	Endif
Return 1
end


david

'$DoRMS|Measure the RMS value of 'visible' Waveform channels in a data file and prints a table of results
'to the log. The user can set the time range to measure and a subset of frames to process. 

	var vh%,ok%,i%,j%;
	var s,e;						'start end time
	var nframes%;				'number of frames
	var sfr%,efr%;				'start, end frame
	var frlist$[3],opt%;
	var chlist%[8];			'list of channels
	var ttl$;					' window title$
	var rms; 					'root mean sqare value in time range
	var tag%;
	var Nignored%:=0;			'count of frames that don't fit the specification

	frlist$[0]:="All"; 	frlist$[1]:="All tagged";	frlist$[2]:="All untagged";
	ToolbarText("Table of rms voltage for visible waveform channels in selected time range");
	View(App(3)).WindowVisible(2);			' iconise the script
	View(Loghandle()).WindowVisible(0);		'hide the log
	vh%:=FileOpen("",0,2,"Select a file to analyse");		'open invisibly; read resource info
	if vh% < 0 then
		halt
	endif;
	
	ttl$:=WindowTitle$();
	Window(0,0,50,50);		'size and show file
	WindowVisible(1);
	nframes%:=FrameCount();
	efr%:=nframes%;
	ChanList(chlist%[],1+2048);		'list of visible waveform channels
	if CursorExists(1) and CursorExists(2) then
		CursorRenumber();
		s:=Cursor(1);
		e:=Cursor(2);
	else
		s:=XLow();
		e:=XHigh();
	endif;

	DlgCreate("Set up",70,14);
	DlgReal(1,"Start time (s)",MinTime(),MaxTime());
	DlgReal(2,"End time (s)",MinTime(),MaxTime());
	DlgInteger(3,"Start at frame..",1,nframes%);
	DlgInteger(4, "End at frame",1,nframes%);
	Dlglist(5,"Frames to include",frlist$[]);
	ok%:=DlgShow(s,e,sfr%,efr%,opt%);
	if not ok% then halt endif; 			'cancelled
	if s >=e then Message("Invalid time range. Please try again."); halt endif;	' guard against invalid settings
	if sfr% >= efr% then Message("Invalid frame range. Please try again.");halt endif;

	Printlog("\n\n%s\n\n", ttl$);				'print filename
	Printlog("Start time (s):\t%f\n",s);
	Printlog("End time (s):\t%f\n\n",e);
	Printlog("Frame\t");
	for i%:= 1 to chlist%[0] do
		Printlog("ch %d (%s)\t",chlist%[i%],ChanUnits$(chlist%[i%])); 'print header
	next;
	Printlog("\n\n");

	for i%:= sfr% to efr% do 'for first to last frame in turn do...
		Frame(i%);				'next frame
		tag%:= FrameTag(i%);
		docase
		case opt% = 2 and tag% = 1 then 'frame tagged but untagged only requested; ignore
			Nignored%+=1;
		case opt% =1 and tag% = 0 then 'frame untagged but tagged only requested; ignore it
			Nignored%+=1;
		else
			Printlog("%d\t",i%);					'print frame number
			for j%:= 1 to chlist%[0] do
				rms:=ChanMeasure(chlist%[j%],10,s,e);		'calculate and print rms values of visible waveform channels
				Printlog("%f\t",rms);
			next;
			Printlog("\n");		'newline
		endcase;
	next;
	if Nignored% = FrameCount() then
		Printlog("No frames matching your specification were found.\n");
	endif;
	Printlog("--------------------------------------------------------\n");
	View(LogHandle());
	Window(0,0,50,100);			'show the results
	FrontView(Loghandle());


david

Recent questions  
Q. I regularly use the graphical sequence editor to define a series of stimuli to for my experiments. 

Usually my stimulus sections only differ in small ways such as varying pulse lengths and 
amplitudes so it would be nice to be able to create a ‘default’ sequence section that could be used 
as a template and copied to other sections for modification. 

A. The graphical sequence editor in the latest version of Spike2 now allows copying of the current 
sequence section to a range of other sections. This makes it much easier to prepare a list of 
similar outputs. Simply press the Copy.. button in the graphical editor and then select the 
outputs to copy, and a range of sections to copy to. 

  
    

User Group  

 

If you have any comments about the newsletter format and content, or wish to un-subscribe from the 
mailing list for this newsletter, please notify sales@ced.co.uk 

mailto:sales@ced.co.uk?subject=eNewsletter
MikeAlbu

